

冒泡排序

基础知识

...

$a(1)$

$a(2)$

$a(3)$

$a(n-1)$

$a(n)$

从右往左逐位比较

```
For i = 1 To n - 1
  For j = n To i + 1 Step -1
    If a(j) < a(j - 1) Then
      t = a(j): a(j) = a(j - 1): a(j - 1) = t
    end if
  Next j
Next i
```

原始冒泡排序特点:

1. 每次从末位开始逐一往前比较, 这种排序称冒泡排序
2. 第1轮确定 $a(1)$, ..., 第 i 轮确定数组 $a(i)$ 的值
3. 第1轮参与数组 n 个, 第2轮参与 $n-1$ 个.....
第 $n-1$ 轮 (最后一轮) 参与排序数组为2个
4. 排好后数组呈升序排列
5. 共比较 $n*(n-1)/2$ 次
6. 外循环控制排序轮数 (第几轮, 共几轮等), 由小到大
7. 内循环控制数组比较起始位置
8. 因为内循环变量 j 跟外循环变量 i 的值相关联, 所以内外循环共同控制数组比较次序




图 01 冒泡实例

改写

| 冒泡 (升序排列) |
|--|
| <pre>For i = 1 To n - 1 For j = n To i + 1 Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t End If Next j Next i</pre> |
| 外循环 i=2 To n |
| <pre>For i = 2 To n For j = n To i Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t End If Next j Next i</pre> |

改写:a(j-1)变为 a(j+1)

```
For i = 1 To n - 1
    For j = n - 1 To i Step -1
        If a(j) > a(j + 1) Then
            t = a(j): a(j) = a(j + 1): a(j + 1) = t
        End If
    Next j
Next i
```

降序排列

```
For i = 1 To n - 1
    For j = n To i + 1 Step -1
        If a(j) > a(j - 1) Then          '只要将 a(j) < a(j - 1)换成 a(j) > a(j - 1)
            t = a(j): a(j) = a(j - 1): a(j - 1) = t
        End If
    Next j
Next i
```

外循环由大到小

```
For i = n - 1 To 1 Step -1
    For j = n To n - i + 1 Step -1
        If a(j) < a(j - 1) Then
            t = a(j): a(j) = a(j - 1): a(j - 1) = t
        End If
    Next j
Next i
```

内外循环都用 Do while 语循环改写

```
i = 1
Do while i <= n - 1
    j = n
    Do while j >= i + 1
        If a(j) < a(j - 1) Then
            t = a(j): a(j) = a(j - 1): a(j - 1) = t
        End If
        j = j - 1
    Loop
    i = i + 1
Loop
```

内循环变量由小到大-沉底排序(升序)

```

For i = 1 To n - 1
    For j = 1 To n - i          '内循由小到大，即每次从第一个数组开始往后比较
        If a(j) > a(j + 1) Then  '请注意语句与冒泡的区别化变
            t = a(j): a(j) = a(j + 1): a(j + 1) = t  '请注意语句与冒泡的区别化变
        End If
    Next j
Next i

```

优化

| 优化 1(已经有序)For | 优化 1(已经有序)do |
|---|--|
| <pre> i = 1 For i = 1 To n - 1 flag = False For j = n To i + 1 Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t flag = True End If Next j If flag = False Then Exit For Next i </pre> | <pre> flag = True i = 1: Do while i <= n - 1 And flag = True flag = False For j = n To i + 1 Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t flag = True End If Next j i = i + 1 Loop </pre> |
| 优化 2(前面有序) | 优化(外循环次数减少)改写简化 |
| <pre> Last = 0 For i = 1 To n - 1 For j = n To i + 1 Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t Last = j End If Next j i = Last - 1 Next i </pre> | <pre> flag = True lun = 0 Do while flag = True flag = False For j = n To 2 Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t flag = True End If Next j Loop </pre> |

| 综合优化改写 do | 综合优化 2-last |
|---|---|
| <pre> flag = True i = 1 Do while i <= n - 1 And flag = True flag = False For j = n To i + 1 Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t i = j flag = True End If Next j Loop </pre> | <pre> i = 1 Last = 1: Last0 = 0 Do while i <= n - 1 And Last0 <> Last ' i <= n - 1 该条件可以省略, 效果一样 Last0 = Last i = Last For j = n To i + 1 Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t Last = j End If Next j Loop </pre> |

其他算法

| 奇偶位冒泡排序 | 奇偶位升降序不同分别排序 |
|--|---|
| <pre> For i = 1 To (n - 1) \ 2 For j = n To i * 2 + 1 Step -1 If a(j) < a(j - 2) Then t = a(j): a(j) = a(j - 2): a(j - 2) = t End If Next j Next i </pre> | <pre> For i = 1 To (n - 1) \ 2 k = 1 For j = n To i * 2 + 1 Step -1 If a(j) * k < a(j - 2) * k Then t = a(j): a(j) = a(j - 2): a(j - 2) = t End If k = -k Next j Next i </pre> |
| 去重复冒泡（末尾前移法） | 擂台排序(注意与选择排序的区别) |
| <pre> bottom = n For i = 1 To bottom - 1 For j = bottom To i + 1 Step -1 If a(j) < a(j - 1) Then t = a(j): a(j) = a(j - 1): a(j - 1) = t ElseIf a(j) = a(j - 1) Then a(j) = a(bottom) bottom = bottom - 1 End If Next j Next i </pre> | <pre> For i = 1 To n - 1 'K = i For j = i + 1 To n If a(j) < a(i) Then t = a(j): a(j) = a(i): a(i) = t ' i 可以用 k 来代替 End If Next j Next i </pre> |

